
HP-35s Calculator Program Mathematics 1

Roots of a Quadratic Equation

Programmer: Dr. Bill Hazelton

Date: March, 2008. Version:    1.0 Mnemonic: Quadratic Roots

Line Instruction User Instructions
Q001 LBL  Q   LBL  Q
Q002 CLSTK   CLEAR  5
Q003 FS?  10   FLAGS  3  .0
Q004 GTO  Q008
Q005 SF  1   FLAGS  1  1
Q006 SF  10   FLAGS  1  .0
Q007 GTO  Q009
Q008 CF  1   FLAGS  2  1
Q009 QUADRATIC ROOT (Key in using EQN  RCL Q  RCL U, etc.,)
Q010 PSE   PSE
Q011 ENTER  A (Key in using EQN  RCL E  RCL N, etc.,)
Q012 PSE   PSE
Q013 INPUT  A   INPUT  A
Q014 ENTER  B (Key in using EQN  RCL E  RCL N, etc.,)
Q015 PSE   PSE
Q016 INPUT  B   INPUT  B
Q017 ENTER  C (Key in using EQN  RCL E  RCL N, etc.)
Q018 PSE   PSE
Q019 INPUT  C   INPUT  C
Q020 RCL  B
Q021 x2   x2

Q022 RCL  A
Q023 RCL×  C
Q024 4
Q025 ×
Q026 –
Q027 STO  D   STO  D
Q028 x < 0?   x?0  3  (<)
Q029 GTO  Q054 [to 2 complex roots]
Q030 x = 0?   x?0  6  (=)
Q031 GTO  Q082 [to 1 real root]
Q032 RCL  D [2 real roots]
Q033 √x
Q034 RCL–  B
Q035 RCL÷  A
Q036 2
Q037 ÷
Q038 STO  P   STO  P
Q039 1ST  REAL  ROOT (Key in using EQN  1  RCL S  RCL T SPACE, etc.)

Q040 PSE   PSE
Q041 VIEW  P   VIEW  P
Q042 RCL  D
Q043 √x
Q044 RCL+  B
Q045 RCL÷  A
Q046 2
Q047 ÷
Q048 +/–
Q049 STO  Q   STO  Q
Q050 2ND  REAL  ROOT (Key in using EQN  2  RCL N  RCL D SPACE, etc.)
Q051 PSE   PSE
Q052 VIEW  Q   VIEW  Q
Q053 GTO  Q091 [to end part]
Q054 RCL  D [2 complex roots]
Q055 ABS   ABS
Q056 √x
Q057 i
Q058 ×
Q059 RCL—B
Q060 RCL÷  A
Q061 2
Q062 ÷
Q063 STO  P   STO  P
Q064 1ST  CMPLX  ROOT (Key in using EQN  1  RCL S  RCL T SPACE, etc.)
Q065 PSE   PSE
Q066 VIEW  P   VIEW  P
Q067 RCL  D
Q068 ABS   ABS
Q069 √x
Q070 +/—
Q071 i
Q072 ×
Q073 RCL—B
Q074 RCL÷  A
Q075 2
Q076 ÷
Q077 STO  Q   STO  Q
Q078 2ND  CMPLX  ROOT (Key in using EQN  2  RCL N  RCL D SPACE, etc.)
Q079 PSE   PSE
Q080 VIEW  Q   VIEW  Q
Q081 GTO  Q091 [to end part]
Q082 RCL  B [1 real root]
Q083 +/—
Q084 RCL÷  A
Q085 2

Q086 ÷
Q087 STO  P   STO  P
Q088 SINGLE  ROOT (Key in using EQN  RCL S  RCL I  RCL  N, etc.)
Q089 PSE   PSE
Q090 VIEW  P   VIEW  P
O091 FS?  1   FLAGS  3  1
Q092 CF  10   FLAGS  2  .0
Q093 STOP R/S
Q094 RTN   RTN

Notes

(1) A simple program to calculate the roots of a quadratic equation in standard form:
ax2 + bx + c = 0.

(2) If there is a single real root for the equation, the value in storage location D is
equal to zero, so after showing SINGLE ROOT, then the value in storage
location P (the solution), the program ends.

(3) If there are two real roots, the program displays 1ST REAL ROOT, displays the
first solution (in storage location P), then displays 2ND REAL ROOT, then
shows the second solution (in storage location Q).

(4) If there are two complex roots, the program displays 1ST COMPLX ROOT,
displays the first solution as a complex number (using the complex number
display format selected), then displays 2ND COMPLX ROOT, and then the
second solution as a complex number.

(5) This program sets Flag 10, so that equations can be displayed as prompts. The
program completes by returning Flag 10 to its setting when the program began.

(6) This program may be called by another program. The roots are left on the stack
after the program finishes, regardless of the approach used, as well as in storage
locations P and Q. To simplify returning to the calling routine, omit the STOP at
line Q093. It is there to allow the stand-alone use of the program to avoid
automatically returning to some other program point.

Theory

The solution to a quadratic equation in the form ax2 + bx + c = 0 is:

� 

x = −b ± b2 − 4ac
2a

where  

� 

b2 − 4ac  is known as the discriminant. The discriminant, commonly termed ∆, is stored
in storage location D.

If ∆ is positive, the equation has two real roots. If ∆ = 0, the equation has a single, real root, i.e., it
is a perfect square. If ∆ is negative, the equation has two complex roots. The calculator solves for
all of these situations.

Running the Program

Press  XEQ  Q, then press ENTER.

Program displays  QUADRATIC ROOT  briefly, then  ENTER  A  briefly, then prompts:
A?
Key in the value of a from the equation. Press R/S.

Program displays  ENTER  B  briefly, then prompts:
B?
Key in the value of b from the equation. Press R/S.

Program displays  ENTER  C  briefly, then prompts:
C?
Key in the value of c from the equation. Press R/S.

A. If there are two real roots, the program briefly displays  1ST  REAL  ROOT, then displays:
P=
and the value of one of the roots of the equation.

Press R/S and the calculator briefly displays  2ND  REAL  ROOT, then displays:
Q=
and the value of the other root of the equation.

Press R/S and the calculator resets the various flags and the program ends.

B. If there are two complex roots, the calculator briefly displays  1ST  CMPLX  ROOT, then
displays:
P=
and the value of one of the roots, as a complex number. The value is displayed in the current
calculator mode for displaying complex numbers. This may be set, and is explained below.

Press R/S and the calculator briefly displays  2ND  CMPLX  ROOT, then displays:
Q=
and the value of the other root, as a complex number.

Press R/S and the calculator resets the various flags and the program ends.

C. If there is a single root to the equation, i.e., the equation is a perfect square, the calculator
briefly displays  SINGLE  ROOT, then displays:
P=
and the value of the single root.

Press R/S and the calculator resets the various flags and the program ends.

Complex Number Display

Complex numbers in the HP-35s can be displayed in two ways in RPN mode. These are the
rectangular mode and polar mode. Which display mode is set depends upon the display setting.
To see complex numbers in rectangular form, press    DISPLAY, then 9. The x i y mode
shows numbers as:  25.3456 i –54.3210

The first part is the real part of the number, and the i separates this from the imaginary part,
which is the second part. So the above number would be read as  25.3456 – i54.3210.

The other display mode, polar, is selected by pressing    DISPLAY, then selecting option 10
(scroll up or down to find it). This is shown as  r θ a, and shows first the distance, then the angle
of this direction as an angle in decimal degrees counter-clockwise from the ℜ  (real) axis
(horizontal axis to the right).

The same number as above would be shown as:  59.943060 θ –64.986744

where the 59.943060 is the distance, and the –60.986744 is the angle in decimal degrees.

Complex numbers may be entered in either mode, but are displayed in the selected mode. The
mode can be set or changed at any time, to suit the user’s needs.

Sample Computations
1. Solve 4x2  +  x  +  8  =  0

There are two complex roots for this equation.

Solutions: 1st Complxx Root  P = -0.125 + i1.408678
2nd Complex Root Q = –0.125 – i1.408678

2. Solve 4x2  +  8x  –  6  =  0

There are two real roots for this equation.

Solutions: 1st Root  P = 0.581139
2nd Root Q = –2.581139

3. Solve x2  +  2x  + 1  =  0

There is just one root for this equation.

Solution: Single Root P = –1.00000

Storage Registers Used

A Value of a in equation
B Value of b in equation
C Value of c in equation
D Discriminant, ∆
P Solution of the first root
Q Solution of the second root
Statistical Registers: not used

Labels Used

Label Q Length  =   390 Checksum  =  4A3E

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computation to check proper operation after entry.


